
MLをはじめ、ほとんどの（高階関数のある）関数型言語は、λ計算という計算モデルに

もとづいて設計・開発されている。「λ計算を知らないと関数型言語を使えない」というわけ

ではないが、計算機科学の理論としては常識なので、「大学で情報科学／情報工学を専攻し

た」と言っても恥ずかしくないように、必要最低限のことをやっておこう。

1 型なしλ計算

1.1 λ式の構文

λ計算には様々な拡張があるが、すべての基礎となる、もっとも純粋な λ計算の抽象構文

は以下のように定義される。

e (λ式) ::= x (変数)

| λx.e (λ抽象)

| e1e2 (関数適用)

変数と関数適用の意味はMLと同様である。λx.eは、Objective Camlの fun x -> eと同

じで、「xを受け取り、eの値を返す」関数である。

関数型言語の計算モデルとかいって、関数しかないじゃん！

（整数とか letはどこへいったの？）

と思うかもしれないが、実は整数や letなども、上のような λ式だけで表せる1。たとえば、

let x = e1 in e2

は

(λx.e2)e1

と表せる。どちらも、まず e1が xに代入され、それから e2が実行されるためである。この

ように、λ計算は定義が極端に簡単（構文はたったの 3行）であるにもかかわらず、普通の

プログラミング言語と同等の計算能力があるので、基礎研究などでは非常に便利である。

なお、括弧が多すぎると読みづらいので、λx.(λy.e)という形の式はλx.λy.eと書く。また、

λx.(e1e2)という形の式を λx.e1e2と略す。つまり、関数適用は λ抽象より優先順位が高い。

さらに、(e1e2)e3という形の式は e1e2e3と書く。つまり、関数適用は左結合的である。これ

は 10− 3− 7と書いたら 10− (3− 7)ではなく (10− 3)− 7になるのと同じことである。

1.2 β簡約

構文だけ定義しても、それが何を意味するか定めないと、言語として成立しない。λ式の

意味は様々な方法で定義できるが、ここではもっとも一般的な small-step操作的意味論の一

つを用いることにする。

純粋な λ計算には関数（λ抽象）しか値がなく、後で見るように組や数などもすべて λ抽

象で表される。したがって、すべての計算は関数適用として表現される。

一般に、関数 λx.e1を引数 e2に適用したら、まず xに e2が代入され、それから e1が実行

されるはずである。これを

(λx.e1)e2 → [e2/x]e1

(R-Beta)

という規則により定義される、λ式と λ式の間の二項関係→で表すことにしよう。ただし
[e2/x]e1は、e1の変数 xに e2を代入した式とする。この二項関係のことを λ式の（1ステッ

プの）β簡約といい、

(λx.e1)e2は [e2/x]e1に（1ステップで）β簡約される

1それどころか、およそコンピュータで可能なすべての計算は、λ計算でも表現できる。厳密にいうと、λ計

算は「チューリング完全」である。つまり、チューリング・マシンで可能な計算は、λ計算でも表現できるし、

逆も言える。

などという。

たとえば

(λx.x)y → y

であり、

(λx.λy.x)z → λy.z

である。また、

(λx.x)(λy.y) → λy.y

となる。

問 1. 上の他に（1ステップの）β簡約の例を挙げよ。 2

1.3 部分式の β簡約

では、(λa.b)cdという λ式は簡約できるだろうか。つまり、

(λa.b)cd → e

となるような式 eはあるだろうか2。先の規則 (R-Beta)は、適用される関数が λ抽象の形に

なっていることを要求している。ところが、今は適用される関数がふたたび (λa.b)cという

関数適用の形をしているので、先の規則 (R-Beta)を当てはめることができない。つまり、

(R-Beta)だけでは、どうやっても (λa.b)cdを簡約することができない。これは、直感に反し

ており、おかしい。たとえば、Objective Camlにおける (fun a -> b) c dの意味とも合わ

ない。

この問題を解決するためには、一般に関数適用 e1e2で、適用される関数 e1を簡約する規

則 (R-App1)が必要となる。

e1 → e′1

e1e2 → e′1e2

(R-App1)

この規則があれば、たとえば先の例の λ式も

(λa.b)cd → bd

2ここで (λa.b)cdは、(λa.b)(cd)ではなく ((λa.b)c)dの略記だったことに注意せよ。したがって、(λa.b)cd =

(λa.b)(cd) → bという簡約はありえない。

と簡約できる。なぜならば、

(λa.b)c → b
R-Beta

(λa.b)cd → bd
R-App1

という導出がちゃんと存在するからである。

なお、(R-App1)と同様の理由により、関数適用 e1e2において引数 e2を簡約する規則 (R-

App2)も必要である。

e2 → e′2

e1e2 → e1e
′
2

(R-App2)

問 2. (R-App2)を用いた簡約の例を挙げよ。また、その簡約の導出を書き下せ。 2

注 1. (R-App1)や (R-App2)は、関数適用 e1e2において e1や e2を簡約するために必要であっ

た。では、λ抽象 λx.eにおいて eを簡約する、以下のような規則 (R-Abs)は必要だろうか？

e → e′

λx.e → λx.e′
(R-Abs)

これは実は「場合による」というのが答えである。たとえば、MLのプログラムを普通に実

行する場合は、fun x -> eの eは関数が適用されるまで評価されないので、(R-Abs)のよ

うな簡約は不要である。しかし、コンパイラによる最適化や、関数と関数の等価性を議論す

る場合は、(R-Abs)のような簡約も必要となる。たとえば、λx.(λy.y)xは λx.xに簡約できる

から、前者を後者におきかえてもよい、などである。この入門では (R-Abs)も規則に含めて

考えることにする。 2

1.4 自由変数、束縛変数、α変換（重要ポイント！）

次に以下のような簡約を考えてみる。

(λx.λy.xy)y → ???

規則 (R-Beta)を素朴に使用すると

(λx.λy.xy)y → λy.yy

となってしまうが、これはおかしい。なぜなら、

式 (λx.λy.xy)yにおいて、一つ目・二つ目の yと、三つ目の yは別々の変数である

はずなのに、上のように簡約した式 λy.yyにおいては、同一の変数になってしまっているか

らである。つまり、そもそも上のような簡約は誤っている。

では、どうすれば良いのか？　先の式

(λx.λy.xy)y

において、一つ目と二つ目の yはただの仮引数であるから、別に必ず yという名前である必

要はない。そこで、一つ目と二つ目の yを新しい変数 y′におきかえた式

(λx.λy′.xy′)y

を考える。この式を簡約すると

(λx.λy′.xy′)y → λy′.yy′

となる。

上の y′のように、λ抽象の仮引数となっている変数のことを束縛変数といい、

式 (λx.λy′.xy′)yにおいて、変数 y′は λy′によって束縛されている

などという。逆に、束縛されていない変数のことを自由変数という。たとえば、

式 (λx.λy′.xy′)yにおいて、変数 yは自由である

などという。

一般に、束縛変数の名前をつけかえても、λ式の意味はかわらない（ただし、同一の束縛

変数の名前は同時につけかえる）。たとえば、λx.xxと λy.yyは同一の関数である。このよ

うに束縛変数の名前をつけかえることを α変換という。λ式 e1が e2に α変換できることを

「e1は e2と α同値である」という。

一方で、自由変数の名前をつけかえたら、λ式の意味はかわってしまう。たとえば、λx.y

と λx.zは

# let y = (fun a -> fun b -> a) ;; (* yを適当に定義しておく *)

val y : ’a -> ’b -> ’a = <fun>

# let z = (fun a -> fun b -> b) ;; (* zを適当に定義しておく *)

val z : ’a -> ’b -> ’b = <fun>

# let e1 = (fun x -> y) ;; (* λ x. y *)

val e1 : ’a -> ’b -> ’c -> ’b = <fun>

# let e2 = (fun x -> z) ;; (* λ x. z *)

val e2 : ’a -> ’b -> ’c -> ’c = <fun>

# e1 "xxx" "yyy" "zzz" ;;

- : string = "yyy"

# e2 "xxx" "yyy" "zzz" ;;

- : string = "zzz"

のように振る舞いが異なる。

注 2. 自由変数・束縛変数の概念は、λ計算だけでなく数理論理学一般に共通である。たと

えば、数学でも

f(x) = x + 1

という関数 f と

g(y) = y + 1

という関数 gは（仮引数の変数名 x, yが異なっていても）等しいとみなされるし、
∫ √

1− x2dx

と ∫ √
1− y2dy

は同じ式であるとみなされる。これは、xや yが束縛変数だから、α変換しても数式の意味

が変化しないためである。 2

β簡約の場合だけでなく、一般に代入 [e2/x]e1を行う際は、e1の束縛変数が e2の自由変数

と衝突しないように、e1をα変換する必要がある。このような代入は、e1の抽象構文につい

ての場合わけと再帰により、以下のように定義できる。

[e2/x]y =





e2 (x = yの場合)

y (x 6= yの場合)

[e2/x](λy.e) = λy′.[e2/x]([y′/y]e) (ただし y′は新しい変数とする)

[e2/x](ee′) = ([e2/x]e)([e2/x]e′)

1.5 Objective Camlによる実装

以上の定義にもとづいて、λ計算のインタプリタをObjective Camlで実装してみよう。

type exp = (* λ式を表すデータ型 *)

Var of string (* 変数 *) (* 変数名は文字列であらわすことにする *)

| Abs of string * exp (* λ抽象 *)

| App of exp * exp (* 関数適用 *)

let gensym = (* 新しい変数名を作る補助的な関数 *)

let counter = ref 0 in (* 整数0を持った参照セルcounterを作る *)

fun () -> (* ダミーの引数 ()を受け取ったら… *)

incr counter; (* counterの中身を一つ増やす *)

"g" ^ string_of_int !counter (* g1, g2, g3, ...等の文字列を返す *)

let rec subst e2 x e1 = (* [e2/x]e1を求めて返す *)

match e1 with

Var(y) ->

if x = y then e2 else Var(y)

| Abs(y, e) ->

let y’ = gensym () in

Abs(y’, subst e2 x (subst (Var(y’)) y e))

| App(e, e’) ->

App(subst e2 x e, subst e2 x e’)

let rec step e = (* eを受け取り、e -> e’となる e’のリストを返す *)

match e with

Var(x) -> []

| Abs(x, e0) ->

(* (R-Abs)より *)

List.map

(fun e0’ -> Abs(x, e0’))

(step e0)

| App(e1, e2) ->

(* (R-Beta)より *)

(match e1 with

Abs(x, e0) -> [subst e2 x e0]

| _ -> []) @

(* (R-App1)より *)

List.map

(fun e1’ -> App(e1’, e2))

(step e1) @

(* (R-App2)より *)

List.map

(fun e2’ -> App(e1, e2’))

(step e2)

let rec repeat e = (* 簡約できなくなるまで stepを反復する *)

match step e with

[] -> e

| e’ :: _ -> repeat e’

例:

# let e1 = Abs("x", Var("x")) ;; (* (λ x. x) *)

val e1 : exp = Abs ("x", Var "x")

# step e1 ;;

- : exp list = []

# let e2 = App(e1, e1) ;; (* (λ x. x) (λ x. x) *)

val e2 : exp = App (Abs ("x", Var "x"), Abs ("x", Var "x"))

# step e2 ;;

- : exp list = [Abs ("x", Var "x")]

# let e3 = App(e2, e2) ;; (* ((λx. x) (λx. x)) ((λx. x) (λx. x)) *)

val e3 : exp =

App (App (Abs ("x", Var "x"), Abs ("x", Var "x")),

App (Abs ("x", Var "x"), Abs ("x", Var "x")))

# step e3 ;;

- : exp list =

[App (Abs ("x", Var "x"), App (Abs ("x", Var "x"), Abs ("x", Var "x")));

App (App (Abs ("x", Var "x"), Abs ("x", Var "x")), Abs ("x", Var "x"))]

# repeat e3 ;;

- : exp = Abs ("x", Var "x")

問 3. 他の例を作って試してみよ。特に、step eが長さ 2以上のリストを返す場合について、

repeat eの結果を考察せよ。 2

1.6 簡約結果の一意性、合流性、簡約戦略

先の例からもわかるように、一つの λ式 eに対して、e → e′となる e′が一つとは限らな

い。つまり、0個かもしれないし、2個以上かもしれない。

0個の場合は、それ以上の簡約ができない、つまり計算が停止したことになる。ただし、計

算の停止といっても、エラーによる異常終了かもしれないし、正常終了かもしれない。どち

らであるかは、簡約結果となる λ式の形により定めることが多い。たとえば「関数適用 e1e2

の形で終わってしまったら、計算が中途半端に停止したとみなしてエラー、その他の形なら

ばOK」などである。

2個以上の場合は、複数の簡約が可能である、つまり計算の方法が一意でないことになる。

では、計算の結果も一意ではなくなってしまうのだろうか？　実は、λ計算では

計算の方法は一意でないが、結果は一意である

ことがわかっている。より正確に記述すると、次の通りである。

定理 1 (簡約結果の一意性). 任意の λ式 eについて、もし e →∗ e1 6→かつ e →∗ e2 6→なら
ば、e1 = e2である。 2

ただし、→∗は→の反射的推移的閉包、つまり→を 0回以上合成した関係である。すなわ

ち、一般に e0 →∗ enとは、e0 → e1 → e2 → . . . → en−1 → en (n ≥ 0)となる e1, e2, . . . , en−1

が存在することである。あるいは

e →∗ e

e1 → e2 e2 →∗ e3

e1 →∗ e3

9



という規則により定義することもできる。また、e 6→とは、e → e′となる e′が存在しないこ

とである。

上の定理は、次の性質から導かれる。

定理 2 (合流性). 任意の λ式 eについて、もし e →∗ e1かつ e →∗ e2ならば、e1 →∗ e′かつ

e2 →∗ e′となる e′が存在する。 2

問 4. 定理 2を仮定して、定理 1を証明せよ。 2

簡約結果の一意性や合流性は、もし簡約が停止すれば結果は一意であると言っているに過

ぎず、絶対に簡約が停止するとは主張していない。実際に、たとえば

Ω = (λx.xx)(λx.xx)

とおくと、

Ω → Ω → Ω → . . .

という、無限に停止しない簡約しか存在しない。

さらに、(λx.y)Ωのような式を考えると、

(λx.y)Ω → y

のように停止する簡約と、

(λx.y)Ω → (λx.y)Ω → (λx.y)Ω → . . .

のように停止しない簡約の両方が存在する。

そこで一般に任意の λ式について、できるだけ簡約が停止するような簡約戦略があるとう

れしい。つまり、「ある簡約が停止するならば、Xにしたがった簡約も停止する」というよ

うなX がほしい。そのようなX としては「できるだけ外側および左側から簡約していく」

最外最左簡約がある。たとえば (λx.y)Ωなら、引数Ωよりも (λx.y)Ω全体のほうが外側にあ

るので、後者から簡約する。最外最左簡約が停止しないような λ式は、どのような簡約も停

止しないことがわかっている。このことから、最外最左簡約は正規順序簡約ともいわれる。

先のインタプリタは、関数 repeatにおいて step eの第一要素しか使っていないが、step

の実装をよく見ると、実は最外最左簡約になっている。

問 5. 先のインタプリタを、最外最左簡約ではなくなるように変えてみよ。また、その違い

がわかるような例を示せ。 2

ちなみにMLは関数適用において、まず引数を評価するので正規順序簡約ではない。実

際に、

# let rec loop x = loop x ;; (* 停止しない関数 *)

val loop : ’a -> ’b = <fun>

# let y = (fun z -> z) ;; (* yを適当に定義する *)

val y : ’a -> ’a = <fun>

# (fun x -> y) (loop ()) ;;

Interrupted.

のように、停止しない引数に λx.yを適用しても停止しない。このように、まず引数を簡約

してから関数を適用する簡略戦略を値呼び (call-by-value)という。逆に、引数を簡約しない

で関数を適用する簡略戦略を名前呼び (call-by-name)という。名前呼び簡約を実現している

言語としては、Haskellがある。ただし、実装では必要呼び(call-by-need)という、より効率

のよい方法を使用している。

1.7 λ式による様々なデータの表現

先に「およそコンピュータで可能な計算は、λ計算でも表現できる」と述べた。したがっ

て、そもそも「およそコンピュータで扱えるデータは、λ式でも表せる」はずである。それ

らの例をいくつか見ていこう。

1.7.1 論理値

論理値 true, falseおよび条件式 if e1 then e2 else e3は、

true = λt.λf.t

false = λt.λf.f

if e1 then e2 else e3 = e1e2e3

と表せる。なぜならば、

if true then e2 else e3 = true e2e3

= (λt.λf.t)e2e3

→ (λf.e2)e3

→ e2

かつ

if false then e2 else e3 = false e2e3

= (λt.λf.f)e2e3

→ (λf.f)e3

→ e3

だからである。

より直感的には、次のように考えることもできる。一般に論理値 bについての、もっとも

基本的な演算は条件分岐 if b then e2 else e3である。そこで、この条件分岐の機能を b自

身にもたせる。つまり、then節 e2と else節 e3を受け取って、そのどちらかを返すような

関数により、bを表すことにする。すると、trueは e2と e3を受け取って、e2のほうを返す

関数

λt.λf.t

で表せる。同じように、falseは e2と e3を受け取って、e3のほうを返す関数

λt.λf.f

で表せる。b自体に条件分岐の機能をもたせたので、if式は単に bに then節と else節を渡

すだけで良い。

1.7.2 組

組は

(e1, e2) = λc.ce1e2

match e1 with (f, s) -> e2 = e1(λf.λs.e2)

のように表せる。

直感的には論理値の場合と同様で、組に対する基本演算であるパターンマッチングの機能

を、組そのものに持たせている。つまり、「パターンマッチングの節を表す関数 cを与えられ

たら、それを第一要素 e1と第二要素 e2に適用する」という関数 λc.ce1e2により、組 (e1, e2)

を表している。すると match式は単に、組（を表す関数）をパターンマッチングの節（を表

す関数）に適用するだけでよい。

問 6. 次の簡約が導けることを確認せよ。

match (e1, e2) with (x, y) -> e3 →∗ [e2/y][e1/x]e3

2

1.7.3 自然数

自然数は

0 = λs.λz.z

1 = λs.λz.sz

2 = λs.λz.s(sz)

3 = λs.λz.s(s(sz))

...

と表せる。このような λ式をチャーチ数ないし自然数のチャーチ符号化(Church encoding)

という。

なぜこのように表すのだろうか？　一般に自然数についての、もっとも基本的な演算はルー

プである。ループの初期値を z、一回ごとに適用する関数を sとすると

• 自然数 0に対するループの値は z

• 1に対する値は s(z)

• 2に対する値は s(s(z))

• 3に対する値は s(s(s(z)))

• 以下同様

となる。sや zの値はループによって異なるので、どのような sと zにも適用できるように

λ抽象して、

0 = λs.λz.z

1 = λs.λz.sz

2 = λs.λz.s(sz)

3 = λs.λz.s(s(sz))

...

となる。

こうすると、たとえば足し算m + nは

m + n = λs.λz.ns(msz)

と定義できる。これは次のように考えられる。まず式mszにより、初期値 zに対して、ルー

プ sをm回繰り返す。そのmszを初期値として sを n回繰り返せば、合わせてm + n回 s

を繰り返したことになるはずである。実際に簡約してみると、たとえば

1 + 2 = λs.λz.1s(2sz)

= λs.λz.(λs.λz.sz)s(2sz)

→ λs.λz.(λz.sz)(2sz)

→ λs.λz.s(2sz)

= λs.λz.s((λs.λz.s(sz))sz)

→ λs.λz.s((λz.s(sz))z)

→ λs.λz.s(s(sz))

= 3

となるし、先のインタプリタで試しても

# let one =

Abs("s",

Abs("z",

App(Var("s"), Var("z")))) ;;

val one : exp = Abs ("s", Abs ("z", App (Var "s", Var "z")))

# let two =

Abs("s",

Abs("z",

App(Var("s"), App(Var("s"), Var("z")))));;

val two : exp = Abs ("s", Abs ("z", App (Var "s", App (Var "s", Var "z"))))

# let plus m n =

Abs ("s",

Abs ("z",

App (App (m, Var("s")),

App(App(n, Var("s")), Var("z"))))) ;;

val plus : exp -> exp -> exp = <fun>

# repeat (plus one two) ;;

- : exp =

Abs ("s", Abs ("z", App (Var "s", App (Var "s", App (Var "s", Var "z")))))

となる。

掛け算m× nは

m× n = λs.λz.n(λz′.msz′)z

と定義できる。これは次のように考えられる。「sをm回繰り返す関数」を n回繰り返せば、

sをm× n回繰り返したことになるはずである。この「sをm回繰り返す関数」が λz′.msz′

であり、それを「n回繰り返す」のが n(λz′.msz′)zである。ループが二重になっているので、

外側のループの初期値 zは全体の初期値と同じだが、内側のループの初期値 z′は呼び出され

るごとに変わっていくことに気をつけよ。

問 7. 1 + 2 →∗ 3を確かめたのと同じように、2× 3 →∗ 6となることを、手動・自動（イン

タプリタ）の両方で確認せよ。 2

問 8. m + nやm × nを λ式で表したように、m − nを λ式で表せ。ただしm < nならば

m− n = 0とする。（この問は難しい。もし自力でできたら住井にメールせよ。） 2

1.7.4 再帰関数

再帰関数 g(y) = eは、

fixf = (λx.f(xx))(λx.f(xx))

という λ式を使って

g = fixλg.λy.e

と表せる3。その理由を説明する。

3eの中に gが含まれているかもしれないので、g = λy.eでは定義になっていないことに注意しよう。

まず、任意の λ式 f について、fixf は

fixf = (λx.f(xx))(λx.f(xx))

→ f((λx.f(xx))(λx.f(xx)))

= f(fixf )

と簡約できる。つまり、fixf は f の不動点になっている。

すると、g(y)は

g(y) = (fixλg.λy.e)y

→ (λg.λy.e)(fixλg.λy.e)y

= (λg.λy.e)gy

→ (λy.e)y

→ e

と簡約される4。よって、g(y) = eという定義が実現されている。

2 単純型つきλ計算

前節の λ計算には型の概念がまったくなかったので、たとえば論理値の falseに対し自然

数の 2を足す

false + 2

等という意味のない演算を行っても、

false + 2 = (λt.λf.f) + 2

= (λs.λz.z) + 2

= 0 + 2

→∗ 2

などと簡約できてしまう。このような混乱があってはアセンブリ言語と大差がなく危険で

ある。

4ここでは、自由な gは g = fixλg.λy.e と定義されたメタな変数であり、束縛された gは通常の λ計算の変

数である。引っかかる人は、fixλg.λy.e のかわりに、α同値な fixλg′.λy.[g′/g]e で確認せよ。

また、通常のプログラミング言語のように、自然数や論理値などを λ式ではなくプリミ

ティブな定数として導入しても、もし型がなかったら

if 3 + 7 then 10 else false

のようにナンセンスな式を書くことができてしまう。

そこで、式の種類すなわち型を区別することにより、プログラムの実行（λ式の簡約）以

前に、そのような混乱を防止する方法について考察する。

2.1 型

まず、型 τ の構文を
τ (型) ::= b (基本型)

| τ1 → τ2 (関数型)

と定義する。ただし基本型というのは、論理値の型 boolや自然数の型 natなど、いわば「型

の定数」である。もし基本型がなかったら、

τ (型) ::= τ1 → τ2 (関数型)

と定義される型 τ の集合は空になってしまうことに注意しよう。

2.2 型判定

型 τ の構文を定義したので、次に「式 eが型 τ を持つ」とはどういうことか考えていく。

たとえば、λx.xという式を考える。この式は、引数 xを与えられたら、その xを結果とし

て返す、という関数である。したがって、型 τ の引数を与えられたら、返値の型も τ となる。

よって、

式 λx.xは型 τ → τ を持つ

といえるはずである。これを

` λx.x : τ → τ

と書く。一般に

式 eが型 τ を持つ

という意味の命題を型判定といい、

` e : τ

と表記する。

では、λx.yという式はどうだろうか。この式は先の式と違い、自由変数 yを持つ。一般に

自由変数の値は式の外から与えられるので、その型も式の外から与えられるはずである。そ

こで、

yが型 τ を持つならば、式 λx.yは型 τ ′ → τ を持つ

という型判定

y : τ ` λx.y : τ ′ → τ

が考えられる。一般に、

x1, x2, . . . , xnがそれぞれ型 τ1, τ2, . . . , τnを持つならば、式 eは型 τ を持つ

という型判定を

x1 : τ1, x2 : τ2, . . . , xn : τn ` e : τ

と書く。たとえば、

m : nat, n : nat, plus : nat→ nat→ nat ` plusmn : nat

という型判定が成り立つはずである。ただし、nat → nat → natとは nat → (nat → nat)

のことである。より一般に、→は右結合的である。
型判定において、変数に型を与えるx1 : τ1, x2 : τ2, . . . , xn : τnという部分のことを型環境と

いい、Γや∆などのメタ変数で表記する。型環境は、変数 x1, x2, . . . , xnから型 τ1, τ2, . . . , τn

への写像とみなすことができる。

2.3 型つけ規則

前節では型判定の表記と直感的な意味について説明したが、それだけでは

どのような Γ, e, τ に対して、Γ ` e : τ という型判定が成り立つのか

まだわからない。これをきちんと定義するのが型つけ規則である。

先に定めた型による、純粋な λ式に対する型つけ規則は、次のように与えられる。

Γ(x) = τ

Γ ` x : τ
(T-Var)

Γ, x : τ1 ` e : τ2

Γ ` λx.e : τ1 → τ2

(T-Abs)

Γ ` e1 : τ ′ → τ Γ ` e2 : τ ′

Γ ` e1e2 : τ
(T-App)

規則 (T-Var)は、型環境 Γに x : τ が含まれていたら、Γの下で変数 xは型 τ を持つ、と述

べている。規則 (T-App)は、型環境Γの下で、式 e1が関数型 τ ′ → τ を持ち、式 e2が引数の

型 τ ′を持つならば、関数適用 e1e2は結果の型 τ を持つ、と述べている。

規則 (T-Abs)の意味は次の通りである。型環境Γの下で、関数 λx.eが型 τ1 → τ2を持つた

めには、Γに引数の型 x : τ1をつけくわえた型環境 Γ, x : τ1の下で、関数の本体 eが型 τ2を

持っていれば良い。これは「xとして型 τ1の引数を与えたら、eは型 τ2の結果を返す」とい

う直感を反映している。

例:

s : nat→ nat, z : nat ` s : nat→ nat
T-Var

s : nat→ nat, z : nat ` z : nat
T-Var

s : nat→ nat, z : nat ` sz : nat
T-App

s : nat→ nat ` λz.sz : nat→ nat
T-Abs

` λs.λz.sz : (nat→ nat) → nat→ nat
T-Abs

2.4 安全性

そもそも型は、false+ 5や if 3 + 7 then 10 else falseといったナンセンスなプログラ

ムを拒否するために導入したのであった。では、前節の規則により型つけできたプログラム

は、本当にエラーにならず、ちゃんと実行できるのだろうか。

そのような性質を証明するためには、まず「エラーとは何か」ということを定めなければ

いけない。λ計算では、プログラムは λ式として表され、その簡約がプログラムの実行に相

当するのであった。したがって、簡約が停止したときに、その λ式がちゃんと何らかの値に

なっていれば正常終了、そうでなければエラーとみなすことができる。さらに純粋な λ計算

では、すべての値は λ抽象すなわち関数として表されるので、「値 = 関数」すなわち

v (値) ::= λx.e (関数)

と定義できる（もし自然数や論理値などをプリミティブな定数として追加したら、それらも

値に含まれる）。

すると、以下の定理が成立する。つまり、前節の規則により型つけできる λ式は、決して

エラーを起こさない、といえる。

定理 3 (安全性). もし ` e : τ かつ e →∗ e′ 6→ならば、e′は値である。 2

この定理は、以下の二つの補題から導かれる。

補題 1 (進行). もし ` e : τ ならば、eは値であるか、あるいは e → e′となる e′が存在する。

2

補題 2 (型の保存). もし ` e : τ かつ e → e′ならば、` e′ : τ である。 2

どちらも証明は ` e : τ の導出についての帰納法による。ただし後者の証明の途中で、以下

の補題が必要となる。

補題 3 (代入補題). もし Γ ` e1 : τ1かつ Γ, x : τ1 ` e2 : τ2ならば、Γ ` [e1/x]e2 : τ2である。

2

証明. Γ, x : τ1 ` e2 : τ2の導出についての帰納法による。 2

問 9. 補題 1と補題 2を利用して、定理 3を証明せよ。 2

2.5 データ抽象

1.7.1節で、論理値および条件式は

true = λt.λf.t

false = λt.λf.f

if e1 then e2 else e3 = e1e2e3

のように λ式で表せると述べた。しかし、論理値・条件式の表し方はこれだけではない。た

とえば、

true = λf.λt.t

false = λf.λt.f

if e1 then e2 else e3 = e1e3e2

のように、trueと falseを逆にしても同じのはずである。したがって、たとえば

true : bool , false : bool , if : bool→ nat→ nat→ nat ` e : nat

のように true, false, ifを使用する（純粋な）λ式 eがあったら、

[λt.λf.t/true][λt.λf.f/false][λb.λt.λf.btf/if]e

のように 1番目の実装を代入しても、

[λf.λt.t/true][λf.λt.f/false][λb.λt.λf.bft/if]e

のように 2番目の実装を代入しても、簡約結果の値は変わらないはずである。

このように、型を用いると単に「エラーが起きない」というだけでなく、

データ構造の実装を変更しても、プログラム全体の結果は変化しない

という性質を保証することができる。このような手法のことをデータ抽象という。

データ抽象のポイントは、「データ構造を使用するプログラムが、その具体的な表され方

を知らず、あくまで抽象的な型の下で扱う」ということにある。たとえば上の例でも、もし

プログラム eが trueを関数として適用したら、2つの実装を区別することができてしまうの

で、eの値が変わらないとはいえなくなる。しかし規則 (T-App)より、そのような関数適用

は型つけできないので、ありえない。

Objective Camlにも、様々な抽象データ構造のライブラリがある。たとえば、値から値へ

の対応を記憶するハッシュテーブルなどである。

# let ht = Hashtbl.create 10 ;; (* 新しいハッシュテーブル htを作る *)

val ht : (’_a, ’_b) Hashtbl.t = <abstr>

# ht ;; (* htの値は抽象化されており見ることができない *)

- : (string, int) Hashtbl.t = <abstr>

# Hashtbl.add ht "abc" 123 ;; (* 値"abc"から値123への対応をhtに追加する *)

- : unit = ()

# Hashtbl.add ht "de" 456 ;; (* "de"から 456への対応を追加する *)

- : unit = ()

# Hashtbl.add ht "f" 789 ;; (* "f"から 789への対応を追加する *)

- : unit = ()

# Hashtbl.find ht "de" ;; (* "de"に対応する値を検索する *)

- : int = 456

# Hashtbl.find ht "g" ;; (* "g"に対応する値は存在しないので例外が発生する *)

Exception: Not_found.

# ht + 1 ;; (* Hashtblモジュールで定義された演算以外は htに適用できない *)

Characters 0-2:

ht + 1 ;; (* Hashtblモジュールで定義された演算以外は htに適用できない *)

^^

This expression has type (string, int) Hashtbl.t but is here used with type

int

さらに勉強するには

λ計算に関する日本語の本は限られているが、たとえば

• プログラミング言語の基礎理論情報数学講座9,大堀淳,共立出版, ISBN 4-320-02659-4.

が良い。英語で良ければ5いろいろとあるが、

• Types and Programming Languages. Benjamin C. Pierce. The MIT Press. ISBN

0-262-16209-1.

がわかりやすい。

5大学でも企業でも、技術書・専門書ぐらい英語で容易に読めないと困るはずだが…

